Zadatak 1. 
Potrebno je sledeći program za hipotetičku mašinu statički rekompajlovati u x86 kod. Pri tome, usvajamo sledeća preslikavanja:

	Hipo mašina
	x86 kod

	akumulator
	al

	X
	cl

	SP
	SP

	PC
	IP

	flegovi: C,Z,P
	FLAGS: C,Z, negirani S

	
	

	
	


           BEG                  ; Count the bits in a num

           LDI 127              ; A=127
   LOOP                         ; REPEAT

           SHR                  ;  A := A DIV 2

           BCC     EVEN         ;  IF A MOD 2 # 0 THEN

           STA     TEMP         ;    TEMP := A

           LDA     BITS

           INC

           STA     BITS         ;    BITS := BITS + 1

           LDA     TEMP         ;    A := TEMP

   EVEN    BNZ     LOOP         ; UNTIL A = 0

           LDA     BITS         ;

           OTI                  ; Write(BITS)

           HLT                  ; terminate execution

   TEMP    DS      1            ; VAR TEMP : BYTE

   BITS    DC      0            ;     BITS : BYTE

           END

Rešenje

Jedna po jedna instrukcija hipo mašine se prevodi odgovarajućom x86 instrukcijom ili nizom instrukcija. Međutim, instrukcije se ne prevode potpuno izolovano jedna od druge, nego je moguće izvršiti npr. optimizaciju ažuriranja flegova.

.386

.model flat, stdcall
INCLUDELIB kernel32.lib
STD_OUTPUT_HANDLE equ -11

GetStdHandle proto :dword
WriteConsoleA proto :dword, :dword, :dword, :dword, :dword
ExitProcess proto :dword
.data 

  temp db 0

  bits db 0

.code

put_char PROC C znak:DWORD
          push eax

          push ebx

          push ecx

          sub esp, 4

          INVOKE GetStdHandle, STD_OUTPUT_HANDLE

          mov ebx, esp

          mov ecx, ebp

          add ecx, 8

          INVOKE WriteConsoleA, eax, ecx, 1, ebx, 0
          add esp, 4

          pop ecx

          pop ebx

          pop eax

          ret

put_char ENDP
main: mov al,127

; LDI 127 kod hipo mašine
lop:    shr al,1


; SHR 
          jnc even1

; BCC EVEN
          mov temp,al

; STA TEMP
          mov al, bits

; LDA BITS
          inc al


; INC
          mov bits,al

; STA BITS
          mov al, temp

; LDA TEMP 
even1:  test al,al

; prethodni mov ne menja flegove, pa mora test da ažurira Z 
          jnz lop


; BNZ LOOP
         ; prevod Write(BITS), znamo da je u pitanju jednocifren broj, inače bi bilo komplik
          mov dl, bits

; DL := bits (cifra za ispis) 
          add dl,30h

; DL := DL + 30h  (u DL ide ASCII kod cifre za ispis)
          push edx
          call put_char

; funkcija koja će ispisati jedan znak na standardni izlaz
          add esp, 4

          ; prevod HLT

          INVOKE ExitProcess, 0
end main

Zadatak 2. 

a) Za zadatu sekvencu koda za hipotetičku mašinu navesti izgled ekvivalentne sekvence x86 koda, ako se prevođenje vrši naredbu po naredbu bez optimizacije ALU indikatora.

b) Isto to, ali sa optimizacijom ALU indikatora, da se računa vrednost samo onih indikatora koji će biti upotrebljeni u kodu.

LOOP1

...
           STA     TEMP         ;    TEMP := A

           LDA     BITS

           INC

           STA     BITS         ;    BITS := BITS + 1

           LDA     TEMP         ;    A := TEMP

           BNZ     LOOP1         ; UNTIL A = 0

Rešenje

a) Koristimo preslikavanja kao u prethodnom zadatku. Želimo da se indikatori C,P i Z hipotetičke mašine preslikaju na x86 flagove C, S (Sign) i Z. P i S imaju komplementarno značenje, tako da pri ispitivanju u x86 kodu treba primeniti suprotan uslov u odnosu na hipo mašinu. Dodatni problem je različit uticaj MOV instrukcija kod x86 u odnosu na LDA kod hipo mašine:

hipotetička mašina
	Mnemon
	C
	PZ
	Hex
	Opis

	LDA  B
	
	*
	19h
	Punjenje A sadržajem memorijske lokacije sa adresom B 

	STA  B
	
	
	1Eh
	Upis vrednosti akumulatora A u memorijsku lokaciju sa adresom B

	INC
	
	*
	05h
	Uvećavanje akumulatora A za 1


MOV instrukcija kod x86 ne utiče na flagove, tako da moramo da upotrebimo dodatnu test instrukciju da bismo ažurirali x86 Z i S flagove. Problem je što test briše C flag, tako da moramo i o tome da vodimo računa. Dakle, ako želimo da se flegovi postave na tačne vrednosti, tada prevod jedne instrukcije

LDA B

glasi


mov al, B


jc CSET


test al, al


jmp KRAJ

CSET
test al, al


stc

KRAJ
...
Postoji i efikasnije rešenje ovog problema, ako uočimo da INC i DEC instrukcije kod x86 utiču na Z i S flegove, ali ne i na C flag. Prevod:

LDA B

glasi


mov al, B


inc al


dec al

Koristeći drugu varijantu, prevod zadate hipo sekvence na x86 asembleru glasi:
loop1: 
...


mov temp, al


mov al, bits


inc al


dec al


inc al


mov bits, al


mov al, temp


inc al


dec al


jnz loop1

Diskusija: Izolovano prevođenje instrukciju po instrukciju odgovara najjednostavnijoj

rekompilaciji npr. u kontekstu JIT kompajlera, ali se dobija neoptimizovan kod.

b) Uočavamo da je potreban samo Z fleg u instrukciji neposredno pre skoka, tako da u ostalim MOV i INC instrukcijama ne ažuriramo flegove, niti vodimo računa o C flagu posle test, čime se dobija tri instrukcije kraće rešenje:
loop1: 
...
          mov temp,al

          mov al, bits

          inc al

          mov bits,al

          mov al, temp

          test al,al

          jnz lop

Zadatak 3

Želimo da emuliramo memoriju procesora koji poseduje segmentno adresiranje. Adresa generisana u programu Address relativna je u odnosu na početak segmenta koji se čuva u registru  BaseRegister. Dužina segmenta čuva se u registru LimitRegister.
Fizička memorija od adrese 0 do adrese ROMSIZE-1 je ROM, a ostatak je RAM. Napisati funkcije emulatora za pristup memoriji. Adresiranje je na nivou jedne memorijske reči.
Rešenje

void WriteMemory(word Address, word Value) {
    word RealAddress;

    RealAddress = BaseRegister + Address;

    if ((Address < LimitRegister) &&

        (RealAddress > ROMSIZE)) {
        Memory[RealAddress] = Value;

    } else {
        RaiseInterrupt(INT_SEGFAULT);

    }
}
word ReadMemory(word Address) {
    word RealAddress;

    RealAddress = BaseRegister + Address;

    if (Address < LimitRegister) {
        return Memory[RealAddress];

    } else {
        RaiseInterrupt(INT_SEGFAULT);

        return NULL;

    }
}
Zadatak 4

Prikazati svaku promenu stanja u translacionom kešu dinamičkog rekompilatora za dati kod hipotetičke mašine. Usvojiti mapiranja kao u zadatku 1. Početak memorije emulirane mašine simbolički je označen sa MEM. Pretpostaviti da JIT kompajler pri svakom pozivu obrađuje izvorišni kod do prve instrukcije skoka, ili do ranije zapamćenog odredišta skoka.

           

BEG                  ; Count the bits in a num

   MEM+0     

LDI 127              ; A=127

   MEM+2
LOOP                         ; REPEAT

   MEM+2        
SHR                  ;  A := A DIV 2

   MEM+3        
BCC     EVEN         ;  IF A MOD 2 # 0 THEN

   MEM+5        
STA     TEMP         ;    TEMP := A

   MEM+7        
LDA     BITS

   MEM+9        
INC

   MEM+10        
STA     BITS         ;    BITS := BITS + 1

   MEM+12        
LDA     TEMP         ;    A := TEMP

   MEM+14
EVEN    BNZ     LOOP         ; UNTIL A = 0

   MEM+16        
LDA     BITS         ;

   MEM+18        
OTI                  ; Write(BITS)

   MEM+19        
HLT                  ; terminate execution

   MEM+20
TEMP    DS      1            ; VAR TEMP : BYTE

   MEM+21
BITS    DC      0            ;     BITS : BYTE

           

END

Rešenje
JIT algoritam

int (*dyncode)(CPU *cpu);

do {

      entrypoint = JIT_translate(cpu->PC);

      dyncode = (int(*)(CPU*)) entrypoint

      status = (*dyncode)(cpu)

} while (isNotEnd(status));
Inicijalno stanje

Translacioni keš je prazan

	Izvorišna adresa
	Adresa u kešu

	MEM+0
	-


Prva iteracija JIT algoritma -- poziv JIT_translate() za MEM+0
Sekvenca koda hipo mašine:

   MEM+0     

LDI 127              ; A=127

   MEM+2
LOOP                         ; REPEAT

   MEM+2        
SHR                  ;  A := A DIV 2

   MEM+3        
BCC     EVEN         ;  IF A MOD 2 # 0 THEN

pod pretpostavkom da se x86 registri moraju pamtiti u memoriji pri povratku u glavnu petlju emulatora, prevodi se u:

saveflags
db 0

saveal

db 0

savex

db 0

savepc

dw 0

blk0:         
mov al,127

; LDI 127 kod hipo mašine
    

shr al,1


; SHR



jnc even1

; BCC EVEN


lahf


; mov flags into AH



mov saveflags, ah
; save flags to memory



mov saveal, al

; save al to memory


mov savex, cl

; save x to memory 




lea cx, MEM+5
; izvorna adresa na koju se prenosi kontrola toka



call BrancHelper
; neće biti vraćanja na mesto iza call
even1:





lahf


; mov flags into AH



mov saveflags, ah
; save flags to memory



mov saveal, al

; save al to memory 




mov savex, cl

; save x to memory 




lea cx, MEM+14
; izvorna adresa na koju se prenosi kontrola toka



call BranchHelper
; neće biti vraćanja na mesto iza call

BranchHelper: (samo opis funkcije, mora biti realizovana u x86 asm kodu)
1. ako je CX jednako -1, skida okvir sa vrha steka i vraća se u glavnu petlju, vraćajući status za kraj emulacije.

2. poziva tcache_lookup() da mapira CX u adresu u kešu (blkN)

ako je poziv uspešan, koristeći povratnu adresu na steku menja se instrukcija lea cx, MEM+N u jmp blkN i kontrola prenosi na blkN

3. U suprotnom (odredište nije u trans. cacheu)

3.1. sa steka se skida okvir BranchHelper funkcije, čime na vrhu steka ostaje okvir kojim je pozvan prethodno izvršeni blok koda. U tom okviru zapisana povratna adresa je adresa glavne petlje emulatora.

3.2. stanje sačuvano u globalnim promjenljivim čuva u CPU strukturi prosleđenoj pri pozivu prevedenog bloka iz glavne petlje emulatora

3.3. izlazi se iz funkcije, čime se kontrola vraća u glavnu petlju emulatora, vraćajući status kojim govori da emulacija još nije završena.

Ažuriranje mapiranja u translacionom kešu:

	Izvorišna adresa
	Adresa u kešu

	MEM+0
	blk0

	MEM+5
	-

	MEM+14
	-


Emulator sada poziva prevedeni x86 kod počev od blk0.

Druga iteracija -- poziv JIT_translate() za MEM+5 jer nije izvršen skok u BCC
Sekvenca koda hipo mašine (samo do MEM+14 jer je to zabeleženo odredište skoka):

   MEM+5        
STA     TEMP         ;    TEMP := A

   MEM+7        
LDA     BITS

   MEM+9        
INC

   MEM+10        
STA     BITS         ;    BITS := BITS + 1

   MEM+12        
LDA     TEMP         ;    A := TEMP

prevodi se u x86 kod:

blk5:
          
mov al, saveal

; restauracija al, flegovi i x ne moraju jer se





; ne koriste u ovom bloku


mov MEM+20,al
; STA TEMP
          

mov al, MEM+21
; LDA BITS
          

inc al


; INC
          

mov MEM+21,al
; STA BITS

          

mov al, MEM+20
; LDA TEMP 



test al,al

; ažuriranje Z flega, ostali se ne koriste dalje





; pa se zato ne rekonstruišu svi flegovi (C)


lahf


; mov flags into AH



mov saveflags, ah
; save flags to memory



mov saveal, al

; save al to memory 




mov savex, cl

; save x to memory 




lea cx, MEM+14



call BranchHelper

Ažuriranje mapiranja u translacionom kešu:

	Izvorišna adresa
	Adresa u kešu

	MEM+0
	blk0

	MEM+5
	blk5

	MEM+14
	-


Emulator poziva prevedeni kod od blk5.

Treća iteracija -- poziv JIT_translate() za MEM+14
Sekvenca koda hipo mašine:

   MEM+14
EVEN    BNZ     LOOP         ; UNTIL A = 0

prevodi se u x86 kod:

blk14:

mov al, saveal

; restauracija al (moglo bi i bez ove instrukcije jer





; se akumulator ne koristi u ovom bloku)


mov ah, saveflags
; restauracija flegova




sahf


; mov flags[7:0], ah



jnz lop


; BNZ LOOP ; ne mora čuvanje, ništa nije menjano


lea cx, MEM+16
; kontekst u ovom bloku nije menjan, pa ga nije


call BranchHelper
; potrebno ponovo čuvati
lop:

lea cx, MEM+2



call BranchHelper

Ažuriranje mapiranja u translacionom kešu:

	Izvorišna adresa
	Adresa u kešu

	MEM+0
	blk0

	MEM+2
	-

	MEM+5
	blk5

	MEM+14
	blk14

	MEM+16
	-


Emulator poziva prevedeni kod od blk14.
Četvrta iteracija -- poziv JIT_translate() za MEM+2
Sada se vidi da od MEM+0 do MEM+5 nije kontinualan osnovni blok. Sekvenca koda hipo mašine:

   MEM+2
LOOP                         ; REPEAT

   MEM+2        
SHR                  ;  A := A DIV 2

   MEM+3        
BCC     EVEN         ;  IF A MOD 2 # 0 THEN

prevodi se u x86 kod:

blk2:         
mov al, saveal

; restauracija al, flegovi ne moraju
    

shr al,1


; SHR



jnc even1

; BCC EVEN


lahf


; mov flags into AH



mov saveflags, ah
; save flags to memory



mov saveal, al

; save al to memory 




jmp blk5

; već imamo prevedeno odredište skoka
even1:





lahf


; mov flags into AH



mov saveflags, ah
; save flags to memory



mov saveal, al

; save al to memory 




jmp blk14

; već imamo prevedeno odredište skoka
Ažuriranje mapiranja u translacionom kešu:

	Izvorišna adresa
	Adresa u kešu

	MEM+0
	blk0

	MEM+2
	blk2

	MEM+5
	blk5

	MEM+14
	blk14

	MEM+16
	-


Emulator poziva prevedeni kod od blk2.

Skok na blk5

Kada se ponovo izvrši skok na blk5, odgovarajući kod se menja:

blk5:
          
mov al, saveal

; restauracija al, flegovi ne moraju



mov MEM+20,al
; STA TEMP

          

mov al, MEM+21
; LDA BITS

          

inc al


; INC

          

mov MEM+21,al
; STA BITS

          

mov al, MEM+20
; LDA TEMP 



test al,al

; ažuriranje Z flega



lahf


; mov flags into AH



mov saveflags, ah
; save flags to memory



mov saveal, al

; save al to memory 




jmp blk14

; OVDE JE PROMENJENO, zna se gde se skače
Skok na blk14
Kada se ponovo izvrši skok na blk14, odgovarajući kod se menja:

blk14:

mov al, saveal

; restauracija al



mov ah, saveflags
; restauracija flegova




sahf


; mov flags[7:0], ah



jnz lop


; BNZ LOOP ; ne mora čuvanje, ništa nije menjano



lea cx, MEM+16



call BranchHelper
lop:

jmp blk2

; OVDE JE PROMENJENO, zna se gde se skače

Peta iteracija glavne petlje -- poziv JIT_translate() za MEM+16
Više puta će se izvršiti segment koda od blk2 do blk14 bez vraćanja u glavnu petlju emulatora, sve dok se ne istisnu svi bitovi iz akumulatora, tj. uslov jnz lop ne bude zadovoljeno. Tada se sekvenca koda hipo mašine:

   MEM+16        
LDA     BITS         ;

   MEM+18        
OTI                  ; Write(BITS)

   MEM+19        
HLT                  ; terminate execution

prevodi se u x86 kod:

blk16:



; nije potrebna restauracija al niti flags
          mov dl, MEM+21
; DL := bits (cifra za ispis) 
          add dl,30h

; DL := DL + 30h  (u DL ide ASCII kod cifre za ispis)

          push EDX


          call put_char

; funkcija iz zadatka 1.
          add ESP, 4
          ; prevod HLT

          mov CX, -1
          call BranchHelper

Ažuriranje mapiranja u translacionom kešu:

	Izvorišna adresa
	Adresa u kešu

	MEM+0
	blk0

	MEM+2
	blk2

	MEM+5
	blk5

	MEM+14
	blk14

	MEM+16
	blk16


Po izvršavanju dela koda od blk16, emulator završava rad.

U odnosu na prikazanu verziju moguće su i dodatne optimizacije. Tako bi instrukcije bezuslovnog skoka koje BranchHelper postavlja pred kraj bloka, umesto lea instrukcije, mogle da se postave umesto prve instrukcije koja služi za čuvanje konteksta emulirane mašine. U tom slučaju bi instrukcija bezuslovnog skoka trebala da skoči na prvu instrukciju posle dela koda koji na početku bazičnog bloka restaurira kontekst emulirane mašine.
